COMMISSION OF THE EUROPEAN COMMUNITIES

energy

Solog — 1
An automatic data-logging system for the concentrated solar irradiation facility (Cosif)

1977

EUR 5643 e

COMMISSION OF THE EUROPEAN COMMUNITIES

energy

Solog — 1
An automatic data-logging system for the concentrated solar irradiation facility (Cosif)

G. BLAESSER, K.H. KREBS

Joint Nuclear Research Centre
Ispra Establishment - Italy
Physics Division

1977

EUR 5643 e

Published by the

COMMISSION OF THE EUROPEAN COMMUNITIES
Directorate-General
'Scientific and Technical Information
and Information Management'
Bâtiment Jean Monnet - Kirchberg

LUXEMBOURG

Printed in Belgium

ABSTRACT

The economic evaluation of the prospects of solar energy utilization depends critically on the availability of reliable "solar" data over extended periods of time. Due to the statistical nature of the problem these data should comprise simultaneously input and output measurements on various conversion systems. In the present report we describe a data logging system which has been developed for this purpose and which is in use with the Concentrated Solar Irradiation Facility (COSIF) at Ispra. After a brief discussion of the general layout we present in detail the electronics of the data logging system; this part is followed by a summary of the corresponding computer program which itself is given in the Appendix, together with the instructions how to set the computer switch board for various data handling options.

CONTENTS

1. INTRODUCTION

The economic utilization of solar energy is mainly conditioned by two factors: the existence of suitable energy converters and the availability of sufficient radiation. Information on the second point can only be obtained by long-term measurements on any particular site. This information becomes even more valuable, if at the same time not only the input but also the output of typical converter systems can be recorded. Future applications will benefit from simple low cost recording systems, as they would allow to check many prospective sites - an advantage which, due to rapidly varying changes in microclimatic conditions, could have a decisive influence on optimal site selection and also on total investment costs; the latter is due to the fact that a detailed knowledge of the distribution of solar energy at a given site can be used to determine the necessary safety margin for the planned installation, as well as size and layout of the storage system.

Moreover, for advanced systems, e.g. those using selective surfaces and/or concentrators, a rather complete knowledge of the quality of the incident radiation is required before the economic aspects can be settled; thus again we are confronted with the need for permanent and relatively inexpensive recording of solar data which is often even outside the usual meteorological practice.

With these objectives in mind, and as necessary support of a specific program to investigate impact and feasibility of photovoltaic conversion of concentrated sunlight, we decided to develop a solar data collection system which should have the following features:

- automatic recording of solar data from more than ten independent input channels,
- storage provisions for unattended collection of hourly data during one week, with the possibility of using shorter integration intervals for a correspondingly shorter measurement period (higher resolution in time),
- visual on-line display,
- hard-copy output on teletype,
- simple design using inexpensive integrated circuits,
- reduced software requirements such that the minicomputer could be replaced later by a simple microprocessor system to be used in a mobile autonomous station.

The present report gives a short description of a first prototype system of this kind, SOLOG-I, its interface electronics and of the PDP-8 computer program, as well as the instructions necessary for its use.

2. SYSTEM OUTLINE

The Concentrated Solar Irradiation Facility (COSIF) has been built to provide comparative data on the input of sunlight and diffused radiation and the corresponding electricity production of photovoltaic converters using different concentrating devices (Fresnel lenses and mirror arrangements) and tracking schemes such as quasistationary (seasonally adjusted) mountings or one-axis and two-axis (heliostatic) mounts (Fig. 1).

The short circuit current (or more generally the current to a constant potential sink) of each converter (consisting of an array of silicon solar cells) is monitored and integrated with respect to time by an integrating operational amplifier connected to a relay-operated digital mAh-counter. In the first phase of the project the readings of the individual counters were manually recorded in a log-book. In principle, these readings should be done after sunset; since this is not always practicable (not even to mention weekends and holidays), the necessity of an automatic recording system became evident. Since no special funds had been foreseen for this installation, it had to be improvised with material available at that time within the physics division. Thus, it was decided to make use of a PDP-8 minicomputer which had been used since 1967 for neutron-physics time-of-flight analysis; due to the close-down of the ISPRA-I reactor in 1973 this computer was available. Initially we tried to use as far as possible also the interface and instrumentation (pulse amplifiers, etc.) of the time-of-flight installation for our data logging purposes. RC-filters in parallel to the counter relays were used to pick up the counting pulses which then were passed via the pulse amplifiers/shapers to the computer. However, this solution turned out to be not at all satisfactory since the pulse amplifiers - constructed for μsec counting - were not able to discriminate sufficiently well between the slow-rising current counter pulses (of duration of 0.1 sec circa) and background noise. Thus it was decided to construct a new set of pulse shaping amplifiers together with a corresponding new interface, all based on available TTL-chips of the 74 series.

This new interface also simplified the writing of the software. On the software part we also imposed simplicity and economy of memory space. Therefore we did not provide teletype control by means of a command language: all commands are inserted directly on the console switch register.

The data (counts corresponding to milli-ampère-hours) are collected for each counter channel in a corresponding storage location in the computer; up to 15 counters can be connected. At preset times ("PRINT INTERVAL") these accumulated counts are transferred to data blocks in memory and the channel counters cleared. Up to 90 such data transfers per channel can be stored.

From the data blocks in the computer memory

- the data can be written out on the teletype machine,
- channel sums can be performed and be written out,
- the data (one channel at a time) can be displayed on an oscilloscope.

The accumulation of counts in a particular channel counter can be monitored from a display on the console lights. Also the running time (minutes since midnight) can be displayed in this way.

3. ELECTRONICS OF THE PULSE SHAPING AND INTERFACE CIRCUITS

The majority of the photovoltaic devices is connected to mAh-counters produced by SOLAREX, Rockville (Md.). The schematic drawing of this counter is shown in Fig. 2. The current from the photovoltaic array passes through the small resistor E (R_E = 0.1 ohm). The resulting voltage drop is amplified by the operational amplifier I and the push-pull transistor pair Tr3, Tr4 which then charges the capacitor C5 in a Miller integrator arrangement. The zero balance of the differential input to the operational amplifier is adjusted by the potentiometer P1.

The charge state of the capacitor C5 is sensed by the second operational amplifier II used as a voltage comparator; if the intervention level determined by the setting of the potentiometer P2 (which can be used to calibrate the counter to exact mA-hour counts by means of a calibrated external constant current source) is reached, the output of the amplifier switches its polarity, i.e. swings from one reference voltage level to the opposite. This switching is amplified in current by Tr5, Tr6 and used to reset (i.e. discharge) C5 via Tr17 to start a new cycle. The switching pulse is also sensed via R14 and C3 by the transistor amplifier Tr1, Tr2, Tr8 which drives the relay of a mechanical digital counter.

For the connection to the computer it was convenient to pick up the pulses at the relay since the pulse amplitude at that point was high enough (20 V) to render negligible the influence of eventual noise sources. However, it was necessary to smoothen the pulse shape, since it showed damped oscillations superimposed on the basic pulse. The pick-up circuit shown in the diagram performs the necessary smoothing and presents at the same time a negligible loading to the counter circuits; in addition to this it acts as dc-isolation.

The outgoing pulse is fed via coaxial cable to the pulse amplifier/shaper shown in Fig. 3, and which consists of a transistor which shifts the level of the pulse to the value necessary for driving the TTL monoflop circuit consisting of one SN 74121. The timing circuit of the monoflop has been chosen such as to obtain standard pulses of a few milliseconds duration. Originally we used standard pulses of the duration of the relay pulse (100 msec), but this unduly increased the dead time of the interface and led to overlap problems when using several counters under high irradiation conditions.

15 such pulse shaping circuits were constructed for up to 15 counters to be connected to the interface. They were mounted in two standard NIM modules, with 8 pulse shapers (channels 1 to 8) in the first crate and 7 (channels 9 to 15) in the second; the second crate contains also a similar shaping circuit for the clock which transforms the output of a 10 Hz crystal controlled oscillator to 1 Hz TTL pulses; it consists of a level-shifting transistor, a monoflop SN 74121 and a divide-by-ten chip SN 7490.

The outputs from the pulse shapers and the clock are connected to the interface shown in Fig. 4, which is mounted in a third NIM module. This interface contains two independent circuits, one for the clock signals and one for the counters, with the corresponding logic devices, device selectors (DS) and level-shifters (D) necessary to go from the TTL-levels (0 and 5 V) to the PDP-8 levels (-3.5 V and 0).

A clock pulse sets the "timer flag" flip-flop (right-hand side of I9) which provokes a program-interrupt (P.I.) signal to the computer. The interrupt routine is written in such a way that the computer determines the origin of the interrupt signal by sequentially questioning the states of the different interrupt devices. The device selector (DS1) of the clock is enabled by the octal combination $(01)_8$ of the "device select"-digits (bits 3 to 8 of an I/O-instruction, i.e. an instruction with PDP-8 instruction code 6). The last three bits (9 to 11) of an I/O-instruction produce independently 3 different pulses (IOP1, IOP2, IOP4). The interface clock circuit sends a "skip" pulse to the computer if it is selected, the timer flag set and an IOP1 pulse received; in this way the instruction $(6011)_8$ causes the computer to skip the next instruction if the timer flag is set. IOP4 is used in combination with the device select code to reset the timer flag. The symbolic expressions for these two instructions are

Timer Sense = TMS = $(6011)_8$

Timer Reset = TMR = $(6014)_8$

A pulse on one of the 15 counter lines (channels) is encoded by the cascaded gates (I1 to I7, consisting alternately of NOR and NAND gates on SN7402 and SN7400 chips, respectively) and recorded as a 4-bit binary channel number on the flip-flops in chips I12 and I13 (SN7473). The setting of these flip-flops is combined with the setting of the "counter flag" flip-flop (left-hand side of I9) which sends an interrupt to the computer. The device select code for the counter section is given by $(11)_8$. IOP2 is used to read the state of the 4-bit flip-flop buffer into the lowest order bits (AC8 to AC11) of the accumulator of the PDP-8, i.e. to read the channel number corresponding to the counter pulse. By the use of the IOP4 pulse both the flag flip-flop and the buffer flip-flops are reset. Thus the software uses the following two instructions for the counter section:

Solar Counter Load = SCL = $(6112)_8$

Solar Counter Reset = SCR = $(6114)_8$

The interface hardware permits also to define a Solar Counter Sense = SCS = $(6111)_8$ which causes a skip of the next instruction if the counter interrupt flip-flop is set, though in the present version of the software the use of this instruction was not necessary.

A manual reset button is provided on the front plate of the interface unit (but not shown in the drawing of Fig. 4); it permits a simultaneous resetting of all flip-flops in both (clock and counter-) sections by drawing all reset lines to zero potential; at start-up, it should be pushed to provide definite initial conditions of the flip-flops.

4. DESCRIPTION OF THE SOFTWARE (PROGRAM SOLE)

The program SOLE occupies 6 memory pages of $(128)_{10}$ = $(200)_8$ positions each.

On page 1 it starts with the low-power interrupt and restart routines, followed by the clock and solar counter routines. The clock routine counts the seconds, resets the seconds counter all 60 seconds while incrementing the minutes counter which at the initiation of the program is set manually to the actual time (minutes past midnight); when reaching 1440 (the number of minutes per day) the minutes counter is reset and the days counter is increased by one. The clock routine also provides for a comparison of the actual time and the "next print time". If the actual time reaches the "next print time" the clock routine on page 1 calls the subroutine OUT1 on page 2 which transfers the contents of the channel memories together with the time reading to the storage memory starting at $(1400)_8$ and then resets the counter memories to zero while computing the "next print time" as the sum of the actual time and the "print interval". The solar counter routine on page 1 causes an increase by one in the double precision (24 bit) memory position identified by the channel number read after the counter-caused interrupt. Physically, the high order and low order bits of these memory positions are stored in separate 12 bit memory cells, the 15 low-order cells starting at KAN on page 6 and the 15 high-order cells consecutively following.

Page 2 of the memory contains the starting procedure, the program for the console display and the subroutine OUT1 which was already discussed. The starting program has different entries that allow either to start a completely new case or to change some of the parameters of the present program. Its use is described by the instructions given in Appendix 1.

The console display program shows the present time (minutes past midnight) in binary form on the console lights if the console switch register setting is zero (all toggle switches down). The binary indications, read in triades as octal numbers, can be converted to hours and minutes by the use of the correspondences given in Table 1. If the switch register setting corresponds to one of the channel numbers, the console lights indicate the contents of the corresponding memory cell (12 low-order bits only). The interrupt option is switched on during display, so that the accumulation of the counts in the different channels continues and can be observed on the console lights.

On page 3 are the subroutines for printing two-digit (TWPRT) and four-digit (CPRT) unsigned decimal numbers and for converting and printing the time in hours and minutes (TPRT). The CPRT-routine is taken from the DEC-standard routines as is the routine UDPRNT on page 4 of the memory, which converts the contents of the two memory cells corresponding to each channel to an unsigned eight-digit decimal number and also printing it in that way. The rest of page 4 is taken by the program VISUAL which displays on the oscilloscope the contents of a number of consecutive memory positions assigned to a channel (which is selected by its number on the switch register).
Interrupt is enabled during VISUAL, so that the data acquisition continues. On page 5 is the write-out program (WRTOUT) which is used to print in one batch all or part of the contents of the memory fields assigned to the storage of the channel readings. This write-out is done "off-line", i.e. the machine does not accumulate any counts during the print-out. The same is true for the program SIGMA which fills page 6 and permits to compute and print partial sums for each channel over specified numbers of consecutive memory positions. The console commands for the programs VISUAL, WRTOUT and SIGMA are summarized in Appendix II. Appendix III contains the listing of the whole program, and Appendix IV is an example of a print-out for a sunny day.

TABLE 1 - Octal Numbers Corresponding to the Different Hours

1 h =	74	13 h =	1414
2 h =	170	14 h =	1510
3 h =	264	15 h =	1604
4 h =	360	16 h =	1700
5 h =	454	17 h =	1774
6 h =	550	18 h =	2070
7 h =	644	19 h =	2164
8 h =	740	20 h =	2260
9 h =	1034	21 h =	2354
10 h =	1130	22 h =	2450
11 h =	1224	23 h =	2544
12 h =	1320	24 h =	2640

APPENDIX I - Program Initiation Console Commands

A. To start a new case (reset everything):

1. Set 200_8 in the switch register (SR)
2. Press LOAD ADDRESS
3. Press START
4. Set TMIN in the SR
5. Press CONTINUE
6. Set TMAX in the SR
7. Press CONTINUE
8. Set PRESENT TIME in the SR
9. Press CONTINUE
10. Set PRINT INTERVAL in the SR
11. Press CONTINUE

B. To change the time reading (adjust the clock):

1. Set 224_8 in the SR
2. Press LOAD ADDRESS
3. Press START
4. Set PRESENT TIME in the SR
5. Press CONTINUE
6. Set PRINT INTERVAL in the SR
7. Press CONTINUE

C. To change only the printing interval:

1. Set 235_8 in the SR
2. Press LOAD ADDRESS
3. Press START
4. Set PRINT INTERVAL in the SR
5. Press CONTINUE

Remarks:

- All times are in minutes past midnight. The octal numbers corresponding to the different hours are given in Table 1,
- TMIN is the time the printing will start in the morning (it must not be zero!),
- TMAX is the time the printing stops in the evening (if it is zero, the printing continues to midnight),
- PRESENT TIME fixes the time reading at the starting moment,
- PRINT INTERVAL is the time between two prints (To the intervals of ten, twenty and thirty minutes correspond the octal numbers 12, 24 and 36),
- While the program is running, its actual time is displayed by the console lights if SR is set to zero, while the content of a channel counter (12 low-order bits) is displayed by the console lights if SR is set to the number of that channel.

APPENDIX II - Output Options Console Commands

A. VISUAL (display on the CRT-scope):

- STOP (and switch CRT on!)
- SR: ∅676
- LOAD ADDRESS
- START (Program stops at 676)
- SR: Number (octal) of display points
- CONTINUE (program stops at 7∅1)
- SR: number (octal) of initial intervals left out (may be ∅)
- CONTINUE (program stops at 7∅5)
- SR: Channel number
- CONTINUE: program shows the memory contents corresponding to the chosen channel on the CRT; a new channel can be chosen simply by changing the corresponding setting of the SR.

B. WRTOUT (print-out of the storage memory)

- STOP
- SR: 1∅∅∅
- LOAD ADDRESS
- START (program stops at 1∅∅∅)
- SR: number of initial intervals left out (may be zero)
- CONTINUE (program prints the contents of the storage, as illustrated in Appendix IV)

C. SIGMA (print-out of sums)

- STOP
- SR: 12∅∅
- LOAD ADDRESS
- START (program stops at 12∅∅)
- SR: number of initial intervals left out (may be zero)
- CONTINUE (program stops at 12∅4)
- SR: number of summands
- CONTINUE (program prints the sums corresponding to the fifteen channels).



Fig. 1 : The Solar Irradiation Facility at the CCR Ispra



Fig. 2 : Schema of the mA-h-counter produced by SOLAREX, Rockville (Md), and the pick-up circuit for the connection of the pulse-shaper



Fig. 3 : Drawing of the pulse amplifier/shaper



Fig. 4: Interface PDP-8

APPENDIX III***PROGRAM LIST:

```
                /SOLE
                /A PROGRAM FOR SOLAR COUNTER CONTROL
                /PART ONE: PROGRAM INTERRUPT ROUTINES
                *0
0000  0000                  0
0001  6102                  SPL             /LOW POWER?
0002  5025                  JMP OTHER       /NO
0003  3113                  DCA AC          /LOW POWER ROUTINE
0004  7010                  RAR
0005  3114                  DCA LINK
0006  1000                  TAD 0
0007  3115                  DCA PC
0010  1013                  TAD RESTRT
0011  3000                  DCA 0
0012  7402                  HLT
0013  5014      RESTRT,     JMP RESUME
0014  7200      RESUME,     CLA
0015  1116                  TAD DOLLR
0016  4517                  JMS I PRINT
0017  4520                  JMS I ABLC
0020  1114                  TAD LINK
0021  7104                  CLL RAL
0022  1113                  TAD AC
0023  6001                  ION
0024  5515                  JMP I PC
0025  3121      OTHER,      DCA SAVEAC
0026  7010                  RAR
0027  3122                  DCA SAVEL
0030  6011                  TMS             /TIMER FLAG SET?
0031  5051                  JMP OTHER1      /NO
0032  6014                  TMR             /RESET TIMER
0033  2123                  ISZ SEC         /INCREMENT SECONDS
0034  5102                  JMP CONTIN
0035  7300                  CLA CLL
0036  1124                  TAD M60
0037  3123                  DCA SEC
0040  2125                  ISZ MINUTE
0041  5071                  JMP EXT
0042  1126                  TAD MINT
0043  3125                  DCA MINUTE
0044  1127                  TAD TLOW
0045  3130                  DCA TCOUNT
0046  2131                  ISZ DAY
0047  5071                  JMP EXT
0050  7402                  HLT
0051  6112      OTHER1,     SCL             /LOAD FROM SOLAR COUNTER BUFFER
0052  6114                  SCR             /AND RESET IT
0053  7450                  SNA             /NO COUNTER?
0054  5110                  JMP OTHER2
0055  3132                  DCA KANAL
0056  1132                  TAD KANAL
0057  1133                  TAD KNO
0060  3135                  DCA KANO
0061  2535                  ISZ I KANO
0062  5102                  JMP CONTIN
```

```
0063  1132           TAD KANAL
0064  1134           TAD KN1
0065  3136           DCA KANO1
0066  2536           ISZ I KANO1
0067  5102           JMP CONTIN
0070  7402           HLT
0071  1126  EXT,     TAD MINT         /NUMBER OF MINUTES PER DAY
0072  7041           CIA              /MINUS NUMBER OF MINUTES
0073  1125           TAD MINUTE       /LEFT FOR THE DAY GIVES THE
0074  3137           DCA MIN          /MINUTES ELAPSED ON THAT DAY
0075  7100           CLL
0076  1130           TAD TCOUNT
0077  1137           TAD MIN          /IF ACTUAL TIME MINUS NEXT
0100  7430           SZL              /PRINT TIME IS NEGATIVE,SKIP,
0101  4540           JMS I OUT        /IF IT IS POSITIVE,PRINT!
0102  7200  CONTIN,  CLA              /END OF COUNTER PROCESSING
0103  1122           TAD SAVEL        /RETURN TO MAIN PROGRAM
0104  7104           CLL RAL
0105  1121           TAD SAVEAC
0106  6001           ION
0107  5400           JMP I 0
0110  6032  OTHER2,  KCC              /IGNORE THE OTHER PERIPHERALS
0111  6042           TCF
0112  5102           JMP CONTIN
0113  0000  AC,      0
0114  0000  LINK,    0
0115  0000  PC,      0
0116  0244  DOLLR,   0244
0117  0513  PRINT,   PRNT
0120  0522  ABLC,    BLC
0121  0000  SAVEAC,  0
0122  0000  SAVEL,   0
0123  0000  SEC,     0
0124  7704  M60,     7704
0125  0000  MINUTE,  0
0126  5140  MINT,    5140
0127  0000  TLOW,    0
0130  0000  TCOUNT,  0
0131  0000  DAY,     0
0132  0000  KANAL,   0
0133  1276  KNO,     KAN
0134  1315  KN1,     KAN+17
0135  0000  KANO,    0
0136  0000  KANO1,   0
0137  0000  MIN,     0
0140  0263  OUT,     OUT1
0141  0000  TCNT,    0
0142  0000  TLIM,    0
0143  0000  UDHIGH,  0
0144  0000  UDLOW,   0
0145  0600  DPRT,    UDPRNT
0146  0535  SPACE,   SPC
0147  7761  K15,     7761
0150  2300  SIZEFD,  2300             /NEGAT.SIZE OF MEMORY FIELD
0151  1400  CASE0,   1400             /BEGIN OF MEMORY FIELD
0152  0000  CNR,     0                /CURRENT MEMORY POINTER NR
0153  0000  CASE,    0                /CURRENT MEMORY ADDRESS
0154  7742  K30,     7742
0155  0000  TW1,     0
0156  0000  CASA,    0
0157  0040  K32,     0040
```

```
                  /SOLE-PART TWO:MAIN PROGRAM AND OUTPUT SUBROUTINE
                  *200
0200   6032                 KCC              /INITIALISATION
0201   6042                 TCF
0202   4520                 JMS I ABLC
0203   1150                 TAD SIZEFD
0204   3152                 DCA CNR
0205   1151                 TAD CASE0
0206   3153                 DCA CASE
0207   1124                 TAD M60
0210   3123                 DCA SEC
0211   3131                 DCA DAY
0212   4326                 JMS INIT
0213   7402                 HLT
0214   7404                 OSR              /READ TMIN
0215   7041                 CIA
0216   3127                 DCA TLOW
0217   7402                 HLT
0220   7404                 OSR              /READ TMAX
0221   7041                 CIA
0222   3142                 DCA TLIM
0223   7000                 NOP
0224   7402                 HLT
0225   7404                 OSR              /READ PRESENT TIME
0226   3137                 DCA MIN
0227   1126                 TAD MINT
0230   1137                 TAD MIN
0231   3125                 DCA MINUTE
0232   1137                 TAD MIN          /INITIAL PRINT TIME
0233   7041                 CIA              /EQUALS PRESENT TIME
0234   3130                 DCA TCOUNT
0235   7402                 HLT
0236   7404                 OSR              /READ PRINT INTERVAL
0237   7041                 CIA
0240   3141                 DCA TCNT
0241   6001                 ION              /END OF INITIALISATION
0242   7300    DISPLAY,     CLA CLL
0243   1261                 TAD R1
0244   3262                 DCA R2
0245   7404                 OSR
0246   7450                 SNA
0247   5257                 JMP DPLTIM
0250   1133                 TAD KNO
0251   3135                 DCA KANO
0252   1535                 TAD I KANO
0253   7000    BRIGHT,      NOP
0254   2262                 ISZ R2
0255   5253                 JMP BRIGHT
0256   5242                 JMP DISPLAY
0257   1137    DPLTIM,      TAD MIN
0260   5253                 JMP BRIGHT
0261   7750    R1,          7750
0262   0000    R2,          0
```

```
0263  0000  OUT1,   0
0264  7300          CLA CLL
0265  1141          TAD TCNT       /INCREMENT PRINT TIME
0266  1130          TAD TCOUNT     /BY THE PRINT INTERVAL
0267  3130          DCA TCOUNT
0270  7100          CLL
0271  1137          TAD MIN        /CHECK IF ACTUAL TIME
0272  1142          TAD TLIM       /IS LARGER THAN TMAX;
0273  7430          SZL            /IF SO, SUPPRESS PRINTING
0274  5315          JMP EXOUT
0275  7300          CLA CLL
0276  1131          TAD DAY
0277  4316          JMS DEPOS
0300  1137          TAD MIN
0301  4316          JMS DEPOS
0302  1154          TAD K30
0303  3155          DCA TW1
0304  1155          TAD TW1
0305  7041          CIA
0306  1133          TAD KNO
0307  3135          DCA KANO
0310  1535          TAD I KANO
0311  4316          JMS DEPOS
0312  2155          ISZ TW1
0313  5304          JMP .-7
0314  4326          JMS INIT
0315  5663  EXOUT,  JMP I OUT1
0316  0000  DEPOS,  0
0317  3553          DCA I CASE
0320  2152          ISZ CNR
0321  7410          SKP
0322  7402          HLT
0323  2153          ISZ CASE
0324  5716          JMP I DEPOS
0325  7402          HLT
0326  0000  INIT,   0              /SUBROUTINE TO RESET FIFTEEN
0327  1147          TAD K15        /COUNTERS AND THEIR EXTENSIONS
0330  3155          DCA TW1
0331  1155  FRST,   TAD TW1
0332  7041          CIA
0333  1133          TAD KNO
0334  3135          DCA KANO
0335  3535          DCA I KANO
0336  1155          TAD TW1
0337  7041          CIA
0340  1134          TAD KN1
0341  3136          DCA KANO1
0342  3536          DCA I KANO1
0343  2155          ISZ TW1
0344  5331          JMP FRST
0345  5726          JMP I INIT
```

```
                /SOLE-PART THREE:PRINT ROUTINES
                *400
0400  0000  TWPRT,    0                    /PRINT WITH OCTAL-
0401  3250            DCA VALUE            /TO-DECIMAL CONVERSION
0402  3251            DCA DIGIT            /FOR TWO-DIGIT NUMBERS
0403  1212            TAD CNTRZ2
0404  3252            DCA CNTRZ
0405  1213            TAD ADDRZ2
0406  3227            DCA ARROW
0407  1200            TAD TWPRT
0410  3214            DCA CPRT
0411  5225            JMP ST2
0412  7776  CNTRZ2,   7776
0413  1260  ADDRZ2,   TAD TENPWR+2
0414  0000  CPRT,     0                    /PRINT WITH OCTAL-TO
0415  3250            DCA VALUE            /DECIMAL CONVERSION FOR
0416  3251            DCA DIGIT            /FOUR-DIGIT NUMBERS
0417  1253            TAD CNTRZ4           /(DIGITAL-8-22-U-SYM)
0420  3252            DCA CNTRZ
0421  1254            TAD ADDRZ4
0422  3227            DCA ARROW
0423  7410            SKP
0424  3250            DCA VALUE
0425  7100  ST2,      CLL
0426  1250            TAD VALUE
0427  1256  ARROW,    TAD TENPWR
0430  7430            SZL
0431  2251            ISZ DIGIT
0432  7430            SZL
0433  5224            JMP ARROW-3
0434  7200            CLA
0435  1251            TAD DIGIT
0436  1255            TAD K260
0437  6046            TLS
0440  6041            TSF
0441  5240            JMP .-1
0442  7200            CLA
0443  3251            DCA DIGIT
0444  2227            ISZ ARROW
0445  2252            ISZ CNTRZ
0446  5226            JMP ARROW-1
0447  5614            JMP I CPRT
0450  0000  VALUE,    0
0451  0000  DIGIT,    0
0452  0000  CNTRZ,    0
0453  7774  CNTRZ4,   7774
0454  1256  ADDRZ4,   TAD TENPWR
0455  0260  K260,     0260
0456  6030  TENPWR,   6030
0457  7634            7634
0460  7766            7766

0461  7777            7777
```

```
0462  0000  TPRT,   0           /SUBROUTINE FOR THE CON-
0463  3307          DCA MIN1    /VERSION OF MINUTES IN
0464  3310          DCA HOUR    /HOURS AND MINUTES WITH
0465  7100  RET,    CLL         /DECIMAL PRINT CONVERSION
0466  1311          TAD HMIN
0467  1307          TAD MIN1
0470  7430          SZL
0471  7410          SKP
0472  5276          JMP WRITE
0473  3307          DCA MIN1
0474  2310          ISZ HOUR
0475  5265          JMP RET
0476  7200  WRITE,  CLA
0477  1310          TAD HOUR
0500  4200          JMS TWPRT
0501  1312          TAD H
0502  4313          JMS PRNT
0503  1307          TAD MIN1
0504  4200          JMS TWPRT
0505  4322          JMS BLC
0506  5662          JMP I TPRT
0507  0000  MIN1,   0
0510  0000  HOUR,   0
0511  7704  HMIN,   7704        /OCT(-60)
0512  0310  H,      0310
0513  0000  PRNT,   0           /BASIC PRINTING ROUTINE
0514  6046          TLS
0515  6041          TSF
0516  5315          JMP .-1
0517  6042          TCF
0520  7200          CLA
0521  5713          JMP I PRNT
0522  0000  BLC,    0           /BELL,CARRIAGE RETURN AND LINE-
0523  1332          TAD BELL    /FEED ROUTINE
0524  4313          JMS PRNT
0525  1333          TAD CART
0526  4313          JMS PRNT
0527  1334          TAD LNFD
0530  4313          JMS PRNT
0531  5722          JMP I BLC
0532  0207  BELL,   0207
0533  0215  CART,   0215
0534  0212  LNFD,   0212
0535  0000  SPC,    0           /SPACE(TWO-BLANKS)PRINT
0536  1343          TAD BLANK
0537  4313          JMS PRNT
0540  1343          TAD BLANK
0541  4313          JMS PRNT
0542  5735          JMP I SPC
0543  0240  BLANK,  0240
```

```
                *600
0600   0000    UDPRNT,    0                  /UNSIGNED DECIMAL PRINT,
0601   7300               CLA CLL            /DOUBLE PRECISION
0602   1245               TAD UDLOOP         /(DIGITAL-8-24-U-SYM)
0603   3250               DCA UDCNT
0604   1246               TAD UDADDR
0605   3255               DCA UDPTR
0606   1655    UDARND,    TAD I UDPTR
0607   2255               ISZ UDPTR
0610   3251               DCA UDHSUB
0611   1655               TAD I UDPTR
0612   2255               ISZ UDPTR
0613   3252               DCA UDLSUB
0614   7100    UDDO,      CLL
0615   1252               TAD UDLSUB
0616   1144               TAD UDLOW
0617   3254               DCA UDTEML
0620   7004               RAL
0621   1251               TAD UDHSUB
0622   1143               TAD UDHIGH
0623   7420               SNL
0624   5232               JMP UDOUT
0625   2253               ISZ UDBOX
0626   3143               DCA UDHIGH
0627   1254               TAD UDTEML
0630   3144               DCA UDLOW
0631   5214               JMP UDDO
0632   7200    UDOUT,     CLA
0633   1253               TAD UDBOX
0634   1247               TAD UDTWO
0635   6046               TLS
0636   6041               TSF
0637   5236               JMP .-1
0640   7300               CLA CLL
0641   3253               DCA UDBOX
0642   2250               ISZ UDCNT
0643   5206               JMP UDARND
0644   5600               JMP I UDPRNT
0645   7770    UDLOOP,    7770
0646   0656    UDADDR,    UDCON1
0647   0260    UDTWO,     0260
0650   0000    UDCNT,     0
0651   0000    UDHSUB,    0
0652   0000    UDLSUB,    0
0653   0000    UDBOX,     0
0654   0000    UDTEML,    0
0655   0000    UDPTR,     0
0656   3166    UDCON1,    3166
0657   4600               4600
0660   7413               7413
0661   6700               6700
0662   7747               7747
0663   4540               4540
0664   7775               7775
0665   4360               4360
0666   7777               7777
0667   6030               6030
0670   7777               7777
0671   7634               7634
0672   7777               7777
0673   7766               7766
0674   7777               7777
0675   7777               7777
```

```
/SCOPE DISPLAY ROUTINE
0676  7402  VISUAL,  HLT
0677  7404           OSR            /NUMBER OF DISPLAY POINTS
0700  3351           DCA NP
0701  7402           HLT
0702  7404           OSR            /DISPLAY STARTING TIME
0703  7040           CMA
0704  3356           DCA MT0
0705  7402           HLT
0706  6001           ION
0707  7300  BEGDPL,  CLA CLL
0710  1356           TAD MT0
0711  3352           DCA MT1
0712  1351           TAD NP
0713  7040           CMA
0714  3353           DCA MNP
0715  3355           DCA N
0716  7404           OSR            /SELECT CHANNEL
0717  7041           CIA
0720  3354           DCA MKD
0721  1151           TAD CASE0
0722  1157  INCR,    TAD K32
0723  2352           ISZ MT1
0724  5322           JMP INCR
0725  1354           TAD MKD
0726  7100  REPETE,  CLL
0727  3156           DCA CASA
0730  1556           TAD I CASA
0731  7000           NOP            /SCALE ON SCREEN CAN BE
0732  7000           NOP            /CHANGED WITH RAL OR RTL
0733  6063           DYL            /INSTEAD OF NOP
0734  7300           CLA CLL
0735  1355           TAD N
0736  7006           RTL
0737  7004           RAL
0740  6057           DXS
0741  2355           ISZ N
0742  2353           ISZ MNP
0743  7410           SKP
0744  5307           JMP BEGDPL
0745  7300           CLA CLL
0746  1156           TAD CASA
0747  1157           TAD K32
0750  5326           JMP REPETE
0751  0000  NP,      0
0752  0000  MT1,     0
0753  0000  MNP,     0
0754  0000  MKD,     0
0755  0000  N,       0
0756  0000  MT0,     0
```

```
1Ø62  ØØØØ  OUTPUT,   Ø
1Ø63  1322            TAD COLMA
1Ø64  3323            DCA COLMB
1Ø65  1147            TAD K15
1Ø66  3155            DCA TW1
1Ø67  1155  RPT,      TAD TW1
1Ø7Ø  7Ø41            CIA
1Ø71  1134            TAD KN1
1Ø72  3136            DCA KANO1
1Ø73  1536            TAD I KANO1
1Ø74  3143            DCA UDHIGH
1Ø75  1155            TAD TW1
1Ø76  7Ø41            CIA
1Ø77  1133            TAD KNO
11ØØ  3135            DCA KANO
11Ø1  1535            TAD I KANO
11Ø2  3144            DCA UDLOW
11Ø3  4545            JMS I DPRT
11Ø4  4546            JMS I SPACE
11Ø5  2323            ISZ  COLMB
11Ø6  5312            JMP CTN
11Ø7  1322            TAD COLMA
111Ø  3323            DCA COLMB
1111  452Ø            JMS I ABLC
1112  2155  CTN,      ISZ TW1
1113  5267            JMP RPT
1114  5662            JMP I OUTPUT
1115  ØØØØ  NSTART,   Ø
1116  ØØØØ  NSTOP,    Ø
1117  ØØØØ  CNR1,     Ø
112Ø  Ø414  CPRT1,    CPRT
1121  Ø462  TPRT1,    TPRT
1122  7772  COLMA,    7772
1123  ØØØØ  COLMB,    Ø
1124  Ø3Ø4  D,        Ø3Ø4
```

```
                /TELETYPE OUTPUT FROM THE MEMORY FIELD
                *1000
1000   7402     WRTOUT,     HLT
1001   7404                 OSR                 /NSTART
1002   7040                 CMA
1003   3315                 DCA NSTART
1004   1152                 TAD CNR
1005   7040                 CMA
1006   1150                 TAD SIZEFD
1007   3317                 DCA CNR1
1010   1151                 TAD CASE0
1011   3156                 DCA CASA
1012   2315     FRST1,      ISZ NSTART
1013   7410                 SKP
1014   5227                 JMP CYCLE
1015   1156                 TAD CASA
1016   1157                 TAD K32
1017   3156                 DCA CASA
1020   7100                 CLL
1021   1317                 TAD CNR1
1022   1157                 TAD K32
1023   7430                 SZL
1024   7402                 HLT
1025   3317                 DCA CNR1
1026   5212                 JMP FRST1
1027   4252     CYCLE,      JMS TRSFER
1030   4720                 JMS I CPRT1         /PRINT THE DAY
1031   1324                 TAD D
1032   4517                 JMS I PRINT
1033   4252                 JMS TRSFER
1034   4721                 JMS I TPRT1         /PRINT THE TIME
1035   1154                 TAD K30
1036   3155                 DCA TW1
1037   1155     REPT1,      TAD TW1
1040   7041                 CIA
1041   1133                 TAD KNO
1042   3135                 DCA KANO
1043   4252                 JMS TRSFER
1044   3535                 DCA I KANO
1045   2155                 ISZ TW1
1046   5237                 JMP REPT1
1047   4262                 JMS OUTPUT
1050   4520                 JMS I ABLC
1051   5227                 JMP CYCLE
1052   0000     TRSFER,     0
1053   1556                 TAD I CASA
1054   2317                 ISZ CNR1
1055   7410                 SKP
1056   7402                 HLT
1057   2156                 ISZ CASA
1060   5652                 JMP I TRSFER
1061   7402                 HLT
```

```
                /CHANNEL SUMS
                *12ØØ
12ØØ   74Ø2     SIGMA,    HLT
12Ø1   74Ø4               OSR
12Ø2   7Ø4Ø               CMA
12Ø3   3272               DCA MTS
12Ø4   74Ø2               HLT
12Ø5   74Ø4               OSR
12Ø6   7Ø41               CIA
12Ø7   3273               DCA NS
121Ø   4667               JMS I INIT1
1211   1151               TAD CASEØ
1212   1157     BEGN,     TAD K32
1213   2272               ISZ MTS
1214   5212               JMP BEGN
1215   327Ø               DCA CASE1
1216   1147               TAD K15
1217   3155               DCA TW1
122Ø   1273     SUM,      TAD NS
1221   3274               DCA NS1
1222   127Ø               TAD CASE1
1223   1155               TAD TW1
1224   3156               DCA CASA
1225   1156               TAD CASA
1226   1147               TAD K15
1227   3275               DCA HCASE
123Ø   1155               TAD TW1
1231   7Ø41               CIA
1232   1133               TAD KNO
1233   3135               DCA KANO
1234   1155               TAD TW1
1235   7Ø41               CIA
1236   1134               TAD KN1
1237   3136               DCA KANO1
124Ø   71ØØ     NSUM,     CLL
1241   1535               TAD I  KANO
1242   1556               TAD I  CASA
1243   3535               DCA I  KANO
1244   743Ø               SZL
1245   7ØØ1               IAC
1246   1536               TAD I  KANO1
1247   1675               TAD I  HCASE
125Ø   3536               DCA I  KANO1
1251   2274               ISZ NS1
1252   526Ø               JMP NEXT
1253   2155               ISZ TW1
1254   522Ø               JMP SUM
1255   4671               JMS I  OUTPT
1256   452Ø               JMS I  ABLC
1257   74Ø2               HLT
```

```
1260  1156  NEXT,    TAD CASA
1261  1157           TAD K32
1262  3156           DCA CASA
1263  1275           TAD HCASE
1264  1157           TAD K32
1265  3275           DCA HCASE
1266  5240           JMP NSUM
1267  0326  INIT1,   INIT
1270  0000  CASE1,   0
1271  1062  OUTPT,   OUTPUT
1272  0000  MTS,     0
1273  0000  NS,      0
1274  0000  NS1,     0
1275  0000  HCASE,   0
1276  0000  KAN,     0


            /END OF PROGRAM SOLE

            /DEFINITIONS FOR THE ASSEMBLER:
            TMS=6011
            TMR=6014
            SCL=6112
            SCR=6114
            SPL=6102
            DYL=6063
            DXS=6057
```

APPENDIX IV: PRINT-OUT OF THE CHANNEL READINGS IN TWO HOUR INTERVALS DURING A SUNNY DAY; ONLY FIVE COUNTERS WERE OPERATING ON THAT DAY.

```
0001D05H00
00000000   00000000   00000000   00000000   00000000   00000000
00000000   00000000   00000000   00000000   00000025   00000000
00000000   00000000   00000000
0001D07H00
00000000   00000000   00000000   00000000   00000000   00000000
00000000   00000000   00000000   00000000   00000129   00000000
00000000   00000203   00000288
0001D09H00
00000000   00000000   00000000   00000000   00000000   00000000
00000000   00000000   00000000   00000126   00000747   00000045
00000000   00000649   00001456
0001D11H00
00000000   00000000   00000000   00000000   00000000   00000000
00000000   00000000   00000001   00000614   00001535   00000261
00000000   00000851   00002485
0001D13H00
00000000   00000000   00000000   00000000   00000000   00000000
00000000   00000000   00000000   00000559   00001417   00000206
00000000   00001039   00002191
0001D15H00
00000000   00000000   00000000   00000000   00000000   00000000
00000000   00000000   00000000   00000400   00001210   00000158
00000000   00001115   00001959
0001D17H00
00000000   00000000   00000000   00000000   00000000   00000000
00000000   00000000   00000000   00000131   00000715   00000055
00000000   00000894   00001333
0001D19H00
00000000   00000000   00000000   00000000   00000000   00000000
00000000   00000000   00000000   00000047   00000362   00000009
00000000   00000443   00000874
0001D21H00
00000000   00000000   00000000   00000000   00000000   00000000
00000000   00000000   00000000   00000000   00000029   00000000
00000000   00000028   00000035
0002D05H00
00000000   00000000   00000000   00000000   00000000   00000000
00000000   00000000   00000000   00000000   00000022   00000000
00000000   00000000   00000000
```

SALES OFFICES

BELGIQUE – BELGIË

Moniteur belge – Belgisch Staatsblad

Rue de Louvain 40-42 – Leuvenseweg 40-42
1000 Bruxelles – 1000 Brussel – Tél. 512 00 26
CCP 000-2005502-27 – Postrekening 000-2005502-27

Sous-dépôt – Agentschap

Librairie européenne – Europese Boekhandel
Rue de la Loi 244 – Wetstraat 244
1040 Bruxelles – 1040 Brussel

DANMARK

J.H. Schultz – Boghandel

Møntergade 19
1116 København K – Tel 14 11 95
Girokonto 1195

BR DEUTSCHLAND

Verlag Bundesanzeiger

Breite Straße
Postfach 108 006
5000 Koln 1 – Tel (0221) 21 03 48
(Fernschreiber Anzeiger Bonn 08 882 595)
Postscheckkonto 834 00 Koln

FRANCE

Service de vente en France des publications des Communautés européennes – Journal officiel

26, rue Desaix – 75 732 Paris Cedex 15
Tél (1) 578 61 39 – CCP Paris 23-96

IRELAND

Stationery Office

Beggar's Bush – Dublin 4
Tel. 68 84 33

ITALIA

Libreria dello Stato

Piazza G. Verdi 10
00198 Roma – Tel (6) 85 08
Telex 62008
CCP 1/2640

Agenzia di Roma

0187 Roma - Via XX Settembre
(Palazzo Ministero del tesoro)

GRAND-DUCHÉ DE LUXEMBOURG

Office des publications officielles des Communautés européennes

5, rue du Commerce
Boîte postale 1003 – Luxembourg
Tél 49 00 81 – CCP 191-90
Compte courant bancaire BIL 8-109/6003/300

NEDERLAND

Staatsdrukkerij- en uitgeverijbedrijf

Christoffel Plantijnstraat, 's Gravenhage
Tel (070) 81 45 11 – Postgiro 42 53 00

UNITED KINGDOM

H M Stationery Office

P.O. Box 569
London SE1 9NH – Tel. 01 - 928 6977, ext. 365
National Giro Account 582-1002

UNITED STATES OF AMERICA

European Community Information Service

2100 M Street, N W – Suite 707
Washington, D.C. 20 037 – Tel. (202) 872 8350

SCHWEIZ–SUISSE–SVIZZERA

Librairie Payot

6, rue Grenus – 1211 Genève
CCP 12-236 Genève - Tel 31 89 50

SVERIGE

Librairie C.E Fritze

2, Fredsgatan – Stockholm 16
Post Giro 193, Bank Giro 73/4015

ESPAÑA

Librería Mundi-Prensa

Castelló 37 – Madrid 1
Tel 275 46 55

OTHER COUNTRIES

Office for Official publications of the European Communities

5, rue du Commerce
Boîte postale 1003 – Luxembourg
Tél 49 00 81 – CCP 191-90
Compte courant bancaire BIL 8-109/6003/300

NOTICE TO THE READER

All scientific and technical reports published by the Commission of the European Communities are announced in the monthly periodical '**euro-abstracts**'. For subscription (1 year : FB 1 200) please write to the address below.

FB 150 DKr 23,60 DM 9,70 FF 20 Lit 3600 Fl 10,10 £ 2.35 US $ 4

OFFICE FOR OFFICIAL PUBLICATIONS OF THE EUROPEAN COMMUNITIES
Boite postale 1003 - Luxembourg

7517